




Zengsongguangdie

赠送光碟

实例教程 详细精解


中文



Visual Basic

精彩编程五十例

王彬华 李建华 编著



电子科技大学出版社

DIANZIKEJIDAXUECHUBANSHE


实例教程 详细精解

中文

Visual Basic

精彩编程五十例

王彬华 李建华 编著



电子科技大学出版社

DIANZIKEJIDAXUECHUBANSHE

图书在版编目 (CIP) 数据

中文 Visual Basic 精彩编程五十例/王彬华, 李建华编著.
—成都: 电子科技大学出版社, 2004.4
ISBN 7-81094-053-8

I.中… II.①王…②李… III.BASIC 语言—程序设计 IV.TP312

中国版本图书馆 CIP 数据核字 (2004) 第 014708 号

内 容 简 介

Visual Basic 是 **Microsoft** 公司开发的、面向对象的以 **BASIC** 语言为基础的编程工具。

本书分类讲述了 **Visual Basic** 的控件使用、界面设计、多媒体应用、系统信息和图形处理、字符串处理、文件操作、数据库编程以及游戏程序等各种应用程序。通过本书的实例导向学习, 让读者能快速地编写应用程序。

本书以实例教程的方式编写, 各部分之间独立性强, 每个实例可以作为一个单独的教程使用。本书不仅适合初学 **Visual Basic** 的读者阅读, 也适用于有 **Visual Basic** 开发经验的读者。由于对各方面的程序开发本书中都有相应的实例, 所以本书可作为一本既全面又实用性较强的 **Visual Basic** 编程及等级考试学习的参考书。

中文 Visual Basic 精彩编程五十例

王彬华 李建华 编著

出 版: 电子科技大学出版社 (成都建设北路二段四号)
策划编辑: 谢应成
责任编辑: 辜守义
发 行: 新华书店经销
印 刷: 四川南方印务有限公司
开 本: 787×1092 1/16 印张 20 字数 486 千字
版 次: 2004 年 4 月第一版
印 次: 2004 年 4 月第一次印刷
书 号: ISBN 7-81094-053-8/TP·255
定 价: 30.00 元 (含光盘)

■ 版权所有 侵权必究 ■

- ◆ 邮购本书请与本社发行科联系。电话: (028) 83201635 邮编: 610054
- ◆ 本书如有缺页、破损、装订错误, 请寄回印刷厂调换。电话: (0833) 7641415



前 言

Visual Basic 是 Microsoft 公司开发的、面向对象的以 BASIC 语言为基础的编程工具。它简单易学，特别适合于编程的初学者以及想要快速创建应用程序的用户。Visual Basic 也是一种伸缩性强的开发工具，可以大大缩短软件产品的开发周期，提高开发效率。

本书没有繁琐的语法介绍，也没有复杂的 Visual Basic 编程环境的介绍。而是由多个实例组成，每个实例都自成体系。本书所有实例代码都经过了认真调试，读者按照其详尽的操作步骤可以快速学会实例制作，从而较快掌握 Visual Basic 编程要领，并从容上手开发实际应用程序。

由于篇幅有限，在实例的归纳注释这一环节略去了一些对学习意义不大的代码。读者在自行编写程序时，可以借助源代码中的部分功能模块，这可以省去很多的时间。

由于时间仓促以及编者水平有限，书中缺点和错误在所难免，恳请广大读者批评指正，并提出宝贵的意见和建议。

编 者

2004 年 4 月



第1例	欢迎使用 Visual Basic.....	1
第2例	命令按钮组.....	4
第3例	列表框的使用.....	6
第4例	计算器.....	10
第5例	加密文本.....	17
第6例	调色板.....	20
第7例	选择字体.....	22
第8例	登录窗口.....	25
第9例	计划任务.....	27
第10例	英文打字练习.....	31
第11例	资源管理器.....	35
第12例	万年历.....	44
第13例	进度条.....	46
第14例	文档编辑器.....	49
第15例	屏幕捕捉.....	54
第16例	托盘程序.....	62
第17例	获取计算机配置信息.....	71
第18例	密码查看器.....	75
第19例	进制转换.....	78
第20例	弹出式菜单.....	88
第21例	位图菜单.....	90
第22例	可以换肤的窗体.....	93
第23例	通用对话框.....	97
第24例	用 FSO 模型建立文件.....	101
第25例	用 FSO 打开文本文件.....	104

第 26 例	多次剪贴板.....	106
第 27 例	用 MMControl 制作的播放器.....	109
第 28 例	图表的制作.....	114
第 29 例	Flash 播放器.....	117
第 30 例	显示声音波形.....	124
第 31 例	清屏效果.....	136
第 32 例	拼图游戏.....	139
第 33 例	弹跳的小球.....	143
第 34 例	百叶窗效果.....	146
第 35 例	雪花飞舞.....	149
第 36 例	屏保程序.....	156
第 37 例	局域网聊天程序.....	163
第 38 例	发送电子邮件.....	168
第 39 例	网页浏览器.....	172
第 40 例	电话拨号程序.....	176
第 41 例	数据库的建立.....	179
第 42 例	编程建立数据库.....	181
第 43 例	使用 ADO 控件操作数据.....	186
第 44 例	数据报表.....	191
第 45 例	用 Word 制作格式合同.....	208
第 46 例	使用 Word 的“艺术字”.....	213
第 47 例	新建 Excel 表格.....	215
第 48 例	将 Excel 数据导入数据库.....	217
第 49 例	扫雷游戏.....	220
第 50 例	学生信息管理系统.....	232

第1例 欢迎使用 Visual Basic

实例说明

本例使用一个简单的标签控件引导读者进入 Visual Basic。该例子让读者熟悉 Visual Basic 编程的基本步骤，了解程序的基本结构。

操作步骤

1. 启动 Visual Basic，以 Windows 98 为例，依次选择“开始”→“程序”→“Microsoft Visual Studio 6.0”→“Microsoft Visual Basic 6.0”打开 Visual Basic 编程环境。
2. 选择菜单“文件”→“新建工程”，弹出如图 1-1 所示对话框。




图 1-1 新建工程

3. 在图 1-1 中选择“标准 EXE”，单击“确定”按钮。Visual Basic 建立一个新工程文件，并自动新建一个窗体，如图 1-2 所示。




图 1-2 窗体

4. 双击窗口左边“工具箱”中的标签 (标签) 控件，在窗体中增加名为 Label1 的

标签控件，用来显示欢迎文本，用鼠标拖动窗体中的 Label1 控件，调整位置使其位于窗体的偏上部位。再用鼠标拖动 Label1 控件周围的 8 个控件点调整控件的大小至适当。

5. 用同样的方法再向窗体中增加一个 CommandButton(命令按钮) 控件，调整其位置和大小至合适。设置其说明属性为“按我试试”。得到如图 1-3 所示界面。




图 1-3 程序界面

6. 双击按钮“按我试试”，调出代码窗口，为按钮的点击(单击)事件编写代码。

7. 选择菜单“运行”→“启动”(或直接按 F5 键)运行程序，单击窗体中的按钮看看程序的运行结果。如果代码输入有错，则会弹出错误提示，将代码进行相应修改即可；如果控件的位置或大小合适，也可运行状态，重新调整其位置和大小，直至合适为止。

8. 将程序修改满意后，还需要编译成可执行文件，使其脱离 Visual Basic 开发环境后也能运行。选择菜单“文件”→“生成工程 1.exe”，在弹出的对话框中为编译的可执行文件输入文件名，单击“确定”按钮 Visual Basic 就开始对我们的程序进行编译。

9. 编译完成后，退出 Visual Basic，直接双击编译的可执行文件看看。

程序代码

本程序的代码非常简单，只为 Command1 命令按钮的点击(单击)事件编写代码，具体程序如下：

```
Private Sub Command1_Click()  
    '设置 Label1 控件的显示内容  
    Label1.Caption="欢迎使用 Visual Basic，将使你步入一个神奇的世界"  
End Sub
```

技术要点

本例主要是让初学者熟悉 Visual Basic 编程环境和程序的基本结构。具体编写一个程序的步骤一般有以下几步(系统分析不在我们的讨论范围之内)：

1. 新建工程和窗体；
2. 在窗体上布置控件，调整其位置和大小；

3. 设置窗体和各控件的相应属性;
4. 编写窗体和各控件的事件代码;
5. 调试程序, 修改其中的错误;
6. 编译生成可执行文件。

在将控件添加进窗体有两种方法, 一种就是前面操作的方法, 在“工具箱”中双击控件, 该控件将放入窗体的中心; 另一种是先用鼠标单击相应控件, 然后在窗体中拖动画出相应的控件。

将控件放入窗体后, 每个控件都变成了应用程序中的对象。我们可以通过设置对象的属性来定制控件, 使其具有不同的外观和作用。要设置属性首先要选中相应对象, 然后在属性窗口中对相应属性进行修改即可。控件的属性和方法非常多, 初学者可先记住一些常用的属性和方法, 然后逐步积累。

第 2 例 命令按钮组

实例说明

命令按钮是我们用得最多的控件之一，其主要作用是接收用户的单击事件。如果一个窗体中有很多命令按钮，分别编写代码费时费力，又不好调试，这时我们可以使用控件数组。本例学习控件数组的用法。

操作步骤

1. 启动 Visual Basic，新建工程。
2. 在窗体中放置一个 CommandButton 控件，其名称为 Command1，单击该控件使其处于选中状态(控件周围有 8 个小黑点)，选择菜单“编辑”→“复制”复制该控件，然后执行粘贴操作。系统弹出如图 2-1 所示提示信息，单击按钮“是”，则在窗体的左上角出现一个名称为 Command1 的控件，这时两个按钮的名称相同，只能通过属性 Index 来区分它们，第一个按钮的 Index 为 0，第二个为 1。



图 2-1 创建控件数组

3. 用同样的方法再粘贴一个按钮，设置三个按钮的 Caption 属性为“按钮 1”、“按钮 2”、“按钮 3”。再在窗体中添加三个 Label 控件，调整其位置和大小如图 2-2 所示。
4. 当按下某个按钮时，就在旁边显示出提示信息。编写代码见后。
5. 调试运行程序，观看看效果如图 2-3 所示。
6. 编译生成可执行文件。




图 2-2 程序界面




图 2-3 运行结果

程序代码

```
Private Sub Command1_Click(Index As Integer)
    Select Case Index + 1
        Case 1
            Label1.Visible = True
            Label1.Caption = "单击了按钮 1"
            Label2.Visible = False
            Label3.Visible = False
        Case 2
            Label2.Visible = True
            Label2.Caption = "单击了按钮 2"
            Label1.Visible = False
            Label3.Visible = False
        Case 3
            Label3.Visible = True
            Label3.Caption = "单击了按钮 3"
            Label1.Visible = False
            Label2.Visible = False
    End Select
End Sub
```

技术要点

在本例中，我们使用了 `Select Case` 语句。Visual Basic 过程能够测试条件表达式，然后根据测试结果执行不同的操作。

因为 Visual Basic 中的数组下标是从 0 开始而不是从 1 开始的。在“属性”窗口中我们可以看到，`CommandButton` 控件数组名为 `Command1(0)`、`Command1(1)` 和 `Command1(2)`。所以在 `Select Case` 语句中要写成 “`Index+1`”，而不是 “`Index`”。

第 3 例 列表框的使用

实例说明

列表框的特点是：列表框中的项目是通过程序插入到其中的，用户无法向清单中输入数据，当选择其中的项目，并在用户单击一个按钮或者执行某个操作时，由应用程序完成对指定项目的具体操作。本例充分说明了列表框的这个特点，并介绍了列表框（ListBox）的常用属性。

操作步骤

1. 新建一个工程文件，设置窗体的名称为 FormListBox，Caption 属性为“列表框应用程序”。

2. 向窗体中添加 8 个 CommandButton 命令按钮控件，设置其属性如表 3-1 所示。

表 3-1 CommandButton 命令按钮控件的属性

控件名称	Caption 属性	控件名称	Caption 属性
List1MoveCmd	删除所选项目	List2MoveCmd	删除所选项目
List1AddCmd	增加新项目	List2AddCmd	增加新项目
List1ClearCmd	清除新项目	List2ClearCmd	清除新项目
LefttoRightCmd	>>	RighttoLeftCmd	<<

3. 向窗体中添加两个 Label 控件，设置其 Caption 属性分别为：未排序的列表、排序列表。

4. 再向窗体中添加两个 ListBox 控件，将 List1 列表框 Sorted 属性设为“False”，List2 列表框 Sorted 属性设为“True”，Style 属性设置为“1-CheckBox”。

5. 调整各控件的位置和大小如图 3-1 所示。

6. 编写相应控件的代码，运行程序。




图 3-1 设计界面

程序代码

Option Explicit

Private Sub Form_Load()

 '装载窗体时，给列表中的元素赋值

 List1.AddItem "China"

 List1.AddItem "America"

 List1.AddItem "Japan"

 List2.AddItem "Soccer"

 List2.AddItem "Tennis"

 List2.AddItem "Basketball"

End Sub

Private Sub LefttoRightCmd_Click()

 '将左边列表中所选元素移动到右边列表中

 If List1.ListIndex >= 0 Then

 '在右边列表中增加新元素

 List2.AddItem List1.Text

 '删除左边列表中的所选元素

 List1.RemoveItem List1.ListIndex

 End If

End Sub

Private Sub List1AddCmd_Click()

 Dim ListItem As String

 ListItem = InputBox("在列表中输入新项目")

 '向列表中加入新项目

 If Trim(ListItem) <> "" Then

 'Trim () 作用是清除 ListItem 字符串内的空格

 List1.AddItem ListItem

 End If

End Sub

Private Sub List1ClearCmd_Click()

 '清空列表

 List1.Clear

End Sub

```
Private Sub List1MoveCmd_Click()  
    Dim i As Integer  
    '判断列表框是否只有一个项目被选中  
    If List1.SelCount = 1 Then  
        List1.RemoveItem List1.ListIndex  
    ElseIf List1.ListCount > 1 Then  
        '删除列表框中所选中的多个项目  
        For i = List1.ListCount - 1 To 0 Step -1  
            'ListCount 返回列表框中的项目总数  
            'ListCount-1 是列表框中最后一个项目的索引号  
            '判断该项目是否被选中, Selected()返回布尔值  
            If List1.Selected(i) Then  
                '删除索引号为 i 的项目  
                List1.RemoveItem i  
            End If  
        Next  
    End If  
End Sub
```

```
Private Sub List2AddCmd_Click()  
    '定义变量, 以接收输入的字符串  
    Dim ListItem As String  
    ListItem = InputBox("在列表中输入新项目")  
    If Trim(ListItem) <> "" Then  
        '将字符串增加到列表中  
        List2.AddItem ListItem  
    End If  
End Sub
```

```
Private Sub List2ClearCmd_Click()  
    '清空列表  
    List2.Clear  
End Sub
```

```
Private Sub List2MoveCmd_Click()  
    Dim i As Integer  
    '判断所选元素是否只有一个  
    If List2.SelCount = 1 Then  
        List2.RemoveItem List2.ListIndex
```

```
    如果所选元素大于 1，则按下列操作进行
ElseIf List2.ListCount > 1 Then
    For i = List2.ListCount - 1 To 0 Step -1
        If List2.Selected(i) Then
            List2.RemoveItem i
        End If
    Next i
End If
End Sub

Private Sub RightToLeftCmd_Click()
    Dim i As Integer
    '判断选中的元素有几个
    If List2.SelCount = 1 Then
        List1.AddItem List2.Text
        List2.RemoveItem List2.ListIndex
        '当选中的元素大于一个时
        ElseIf List2.SelCount > 1 Then
            For i = List2.ListCount - 1 To 0 Step -1
                '如果该元素被选中，则将其从右边移至左边
                If List2.Selected(i) Then
                    List1.AddItem List2.List(i)
                    List2.RemoveItem i
                End If
            Next i
        End If
    End If
End Sub
```

技术要点

在 ListBox 控件中，ListCount 属性返回列表框中的选项总数；ListIndex 返回该控件中选定项目的索引。用 List1.List(List1.Listindex) 可以获取选项的文本。

第4例 计 算 器

实例说明

在 Windows 操作系统的附件中为用户提供了一个计算机器, 本例模仿它设计一个简单的计算器程序, 使读者进一步掌握控件数组的使用。

操作步骤

1. 新建一个工程文件, 修改窗体的 Caption 属性为“计算器”。
2. 向窗体中添加一个 CommandButton 控件, 设置名称为 cmdNumber, 参照第 2 例中的方法, 建立一个 cmdNumber 控件数组, 共十个按钮表示计算器中 10 个数字, 设置 Index 和 Caption 属性为 0~9; 同样再建立一个 CommandButton 控件数组 cmdOperator 来作为运算符的按钮, 设置 Index 为 1~4, Caption 属性为 “+”、“-”、“*” 和 “/”。
3. 再向窗体中添加 5 个按钮, 设置其名称分别为 “cmdPlusMinus”、“cmdDecimal”、“cmdClearEntry”、“cmdClear” 和 “cmdEquals”, Caption 分别为 “+/-”、“.”、“CE”、“C” 和 “=”。
4. 最后向窗体中加入一个 TextBox 控件, 设置其名称为 txtDisplay, 设置其 Text 属性为空。调整各控件的位置, 最后结果如图 4-1 所示。




图 4-1 设计界面

程序代码

```
Option Explicit
```

```
Private StoredValue As Double
```

```
Private Const opNone = 0
```

```
Private Const opAdd = 1
Private Const opSubtract = 2
Private Const opMultiply = 3
Private Const opDivide = 4
Private Operator As Integer
```

```
Private NewEntry As Boolean
```

'删除最后的字符

```
Private Sub DeleteCharacter()
```

```
    Dim txt As String
```

```
    Dim min_len As Integer
```

```
    txt = txtDisplay.Text
```

```
    If Left$(txt, 1) = "-" Then
```

```
        min_len = 2
```

```
    Else
```

```
        min_len = 1
```

```
    End If
```

```
    If Len(txt) > min_len Then
```

```
        txtDisplay.Text = Left$(txt, Len(txt) - 1)
```

```
    Else
```

```
        txtDisplay.Text = "0"
```

```
    End If
```

```
End Sub
```

'清除显示内容，保存运算符

```
Private Sub cmdClear_Click()
```

```
    cmdClearEntry_Click
```

```
    StoredValue = 0
```

```
    Operator = opNone
```

```
End Sub
```

'清除显示内容

```
Private Sub cmdClearEntry_Click()
```

```
    txtDisplay.Text = ""
```

```
End Sub
```